

Classic Pendulum 05-16-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[9]:=

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
  StyleDefinitions -> Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

I use Mathematica to work through some calculations for the classical pendulum. The original notebook was *Pendulum 1 03-26-05*, and was instigated by my reading of *Nonlinear Physics with Maple for Scientists and Engineers* by R. Enns and G. McGuire. I make some extensions of the earlier notebook.

Included below are

- 1 Derivation of basic equation for a classical pendulum from the perspectives of force balance and the Lagrangian framework.
- 2 Solution under the linear approximation
- 3 Attempt at immediate direct solution using Mathematica
- 4 Classic solution via first integrals and manipulations leading to elliptic functions
- 5 Calculation of the dependence of the period on oscillation amplitude and comparison again numerical results

Notational preparations

In the following I make use of mathematical notations involving subscripts and superscripts. Manipulations involving such decorated symbols using Mathematica sometimes encounter difficulties if those quantities are not explicitly declared to be symbols. I use the Notation package “symbolize” some quantities.


In[11]:=

```
<< Notation`;
```

```
In[12]:= Symbolize[ $\theta$ ]; Symbolize[ $\dot{\theta}$ ]; Symbolize[ $y_\theta$ ]; Symbolize[ $\phi_\theta$ ];
```

I Derivation

The problem geometry is



I derive the equation of motion from force balance

```
In[13]:= w1[1] = {m D[x[t], {t, 2}] == -T Sin[theta[t]],
            m D[z[t], {t, 2}] == T Cos[theta[t]] - m g}
```

```
Out[13]:= {m x''[t] == -T Sin[theta[t]], m z''[t] == -g m + T Cos[theta[t] ]}
```

```
In[14]:= w1[2] = w1[1] /. {x -> ((L Sin[theta[#]]) &), z -> ((-L Cos[theta[#]]) &)}
```

```
Out[14]:= {m (-L Sin[theta[t]] e'[t]^2 + L Cos[theta[t]] e''[t]) == -T Sin[theta[t]],
           m (L Cos[theta[t]] e'[t]^2 + L Sin[theta[t]] e''[t]) == -g m + T Cos[theta[t] ]}
```

```
In[15]:= w1[3] = w1[2][[2]] /. Solve[w1[2][[1]], T][[1, 1]] // Simplify
```

```
Out[15]:= m (g + L Csc[theta[t]] e''[t]) == 0
```

```
In[16]:= w1[4] = (# Sin[theta[t]] / (m L)) & /@ w1[3] // ExpandAll
```

```
Out[16]:=  $\frac{g \sin[\theta[t]]}{L} + e''[t] == 0$ 
```

I introduce a symbol for the angular frequency.

In[17]:= **def[ω] = ω == $\sqrt{g/L}$**

Out[17]=
$$\omega == \sqrt{\frac{g}{L}}$$

In[18]:= **w1[5] = w1[4] /. Sol[def[ω], g]**

Out[18]=
$$\omega^2 \text{Sin}[\theta[t]] + \theta''[t] == 0$$

Of course this equation of motion can also be obtained from the Lagrangian framework. The potential energy and kinetic energy are

In[19]:= **w1[6] = { $\mathcal{V} == m g L (1 - \text{Cos}[\theta[t]]$), $\mathcal{T} == \frac{m}{2} D[(L \theta[t])^2, \{t, 2\}]$ }**

Out[19]=
$$\{\mathcal{V} == g L m (1 - \text{Cos}[\theta[t]]), \mathcal{T} == \frac{1}{2} L^2 m (2 \theta'[t]^2 + 2 \theta[t] \theta''[t])\}$$

The Lagrangian is

In[20]:= **w1[7] = $\mathcal{L} == \mathcal{T} - \mathcal{V}$ /. (w1[6] /. Equal \rightarrow Rule)**

Out[20]=
$$\mathcal{L} == -g L m (1 - \text{Cos}[\theta[t]]) + \frac{1}{2} L^2 m (2 \theta'[t]^2 + 2 \theta[t] \theta''[t])$$

The Euler Lagrange equation is

In[21]:= **w1[8] = ((D[D[#], D[$\theta[t]$], t], t] - D[#], $\theta[t]$)) & /@ w1[7][[2]] == 0**

Out[21]=
$$g L m \text{Sin}[\theta[t]] + L^2 m \theta''[t] == 0$$

or

In[22]:= **w1[9] = (# / (m L^2)) & /@ w1[8] /. Sol[def[ω], g] // Simplify**

Out[22]=
$$\omega^2 \text{Sin}[\theta[t]] + \theta''[t] == 0$$

which, of course, is the same equation obtained using force balance.

For convenience I define a version of this equation that is initialized when the notebook is opened.

In[23]:= **w1["final"] = $\omega^2 \text{Sin}[\theta[t]] + \theta''[t] == 0$**

Out[23]=
$$\omega^2 \text{Sin}[\theta[t]] + \theta''[t] == 0$$

2 Approximate solution

In[24]:= `w2[1] = w1["final"]`

Out[24]= $\omega^2 \text{Sin}[\theta[t]] + \theta''[t] == 0$

Noting

In[25]:= `Sin[x] ≈ Normal@Series[Sin[x], {x, 0, 5}]`

Out[25]=
$$\text{Sin}[x] \approx x - \frac{x^3}{6} + \frac{x^5}{120}$$

The linear approximation is

In[26]:= `w2[2] = w2[1] /. Sin[θ[t]] → θ[t]`

Out[26]= $\omega^2 \theta[t] + \theta''[t] == 0$

I solve for initial conditions $\dot{\theta}(0) = 0$, $\theta(0) = \theta_0$ (it's maximum angle of deflection)

In[27]:= `w2[3] = DSolve[{w2[2], θ'[0] == 0, θ[0] == θ0}, θ[t], t][[1, 1]] // Expand // RE`

Out[27]=
$$\theta[t] == \theta_0 \text{Cos}[t \omega] + \frac{\dot{\theta}_0 \text{Sin}[t \omega]}{\omega}$$

3 Towards an exact solution

The pendulum has a rich history in physics and mathematics and was analyzed by several prominent 19th century mathematicians. Wikipedia has a nice introduction to this history — [https://en.wikipedia.org/wiki/Pendulum_\(mathematics\)](https://en.wikipedia.org/wiki/Pendulum_(mathematics)).

I attempt a direct solution using Mathematica.

In[28]:= `w3[1] = w1["final"]`

Out[28]= $\omega^2 \text{Sin}[\theta[t]] + \theta''[t] == 0$

Mathematica has difficulty with the solution of nonlinear equation. The analog of w2[3] does not work.

In[29]:=

```
w3[2] = DSolve[{w3[1],  $\theta'[0] == \dot{\theta}_0$ ,  $\theta[0] == \theta_0$ },  $\theta[t], t$ ]
```

- ... **Solve:** Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.
- ... **DSolve:** For some branches of the general solution, unable to solve the conditions.
- ... **DSolve:** For some branches of the general solution, unable to solve the conditions.

Out[29]=

```
{}
```

However, Mathematica can solve the pendulum equation when initial conditions are are not specified.

In[30]:=

```
w3[3] = DSolve[w3[1],  $\theta[t], t$ ] [[1, 1]] // RE
```

- ... **Solve:** Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[30]=

$$\theta[t] == -2 \text{JacobiAmplitude}\left[\frac{1}{2} \sqrt{\left((2 \omega^2 + C[1]) (t + C[2])^2\right)}, \frac{4 \omega^2}{2 \omega^2 + C[1]}\right]$$

I set up the equations for imposing the initial conditions

In[31]:=

```
w3[4] = (D[#, t] &) /@ w3[3]
```

Out[31]=

$$\theta'[t] == -\left(\left(\left(2 \omega^2 + C[1]\right) (t + C[2]) \text{JacobiDN}\left[\frac{1}{2} \sqrt{\left((2 \omega^2 + C[1]) (t + C[2])^2\right)}, \frac{4 \omega^2}{2 \omega^2 + C[1]}\right]\right) / \left(\sqrt{\left((2 \omega^2 + C[1]) (t + C[2])^2\right)}\right)\right)$$

In[32]:=

```
w3[5] = {w3[3], w3[4]} /. t -> 0 /. { $\theta'[0] \rightarrow \dot{\theta}_0$ ,  $\theta[0] \rightarrow \theta_0$ }
```

Out[32]=

$$\left\{\theta_0 == -2 \text{JacobiAmplitude}\left[\frac{1}{2} \sqrt{\left(2 \omega^2 + C[1]\right) C[2]^2}, \frac{4 \omega^2}{2 \omega^2 + C[1]}\right], \dot{\theta}_0 == -\left(\left(\left(2 \omega^2 + C[1]\right) C[2] \text{JacobiDN}\left[\frac{1}{2} \sqrt{\left(2 \omega^2 + C[1]\right) C[2]^2}, \frac{4 \omega^2}{2 \omega^2 + C[1]}\right]\right) / \left(\sqrt{\left(2 \omega^2 + C[1]\right) C[2]^2}\right)\right)\right\}$$

But, Mathematica cannot immediately solve this nonlinear system involving special functions.

```
In[33]:= w3[6] = Solve[w3[5], {C[1], C[2]}]
```

... Solve: This system cannot be solved with the methods available to Solve.

```
Out[33]:= Solve[{θ = -2 JacobiAmplitude[ $\frac{1}{2} \sqrt{(2 \omega^2 + C[1]) C[2]^2}$ ,  $\frac{4 \omega^2}{2 \omega^2 + C[1]}$ ],
  θ' = -(( (2 ω² + C[1]) C[2] JacobiDN[ $\frac{1}{2} \sqrt{(2 \omega^2 + C[1]) C[2]^2}$ ,  $\frac{4 \omega^2}{2 \omega^2 + C[1]}$ ]) /
  (√((2 ω² + C[1]) C[2]²)))}, {C[1], C[2]}]
```

Certainly someone with expert knowledge of identities for elliptic functions could solve these equations, but the specific tricks required are not immediately available to Mathematica.

4 Solving the nonlinear pendulum

As just observed, the equations describing a pendulum are difficult to solve, even given the capabilities of a modern symbolic computation engine. However, some changes of variable and trigonometric manipulations render them tractable. The following calculations are classic. For example, see the exposition in *Principles of Mechanics*, Sygne and Griffith, 1949, which is available on line.

A first integral is easy to obtain. Start with the general equation from section 1

```
In[34]:= w4[1] = w1["final"]
```

```
Out[34]:= ω² Sin[θ[t]] + θ'[t] == 0
```

Multiplying by $\theta'(t)$.

```
In[35]:= w4[2] = (# θ'[t]) & /@ w4[1] // ExpandAll
```

```
Out[35]:= ω² Sin[θ[t]] θ'[t] + θ'[t] θ''[t] == 0
```

```
In[36]:= w4[3] = MapEqn[Integrate[#, t] &, w4[2]]
```

```
Out[36]:= -ω² Cos[θ[t]] +  $\frac{1}{2}$  θ'[t]² == 0
```

```
In[37]:= w4[4] = Solve[w4[3] /. θ'[t]² → x, x][[1, 1]] /. x → θ'[t]² // RE
```

```
Out[37]:= θ'[t]² == 2 ω² Cos[θ[t]]
```

The constant of integration has to be introduced manually

In[38]:= $w4[5] = w4[4][[1]] == w4[4][[2]] + C$

Out[38]:= $\theta'[t]^2 == C + 2 \omega^2 \cos[\theta[t]]$

Solve for the constant of integration

In[39]:= $w4[6] = w4[5] /. t \rightarrow \theta /. \{\theta'[0] \rightarrow \theta, \theta[0] \rightarrow \theta_0\}$

Out[39]:= $\theta == C + 2 \omega^2 \cos[\theta_0]$

In[40]:= $w4[7] = w4[5] /. \text{Sol}[w4[6], C]$

Out[40]:= $\theta'[t]^2 == -2 \omega^2 \cos[\theta_0] + 2 \omega^2 \cos[\theta[t]]$

Mathematica can solve this ode but the solution is quite messy and involves imaginary arguments of special functions. It is worthwhile to see how this equation was solved in olden days.

In[41]:= $w4[8] = \text{DSolve}[\{\theta'[t]^2 == -2 \omega^2 \cos[\theta_0] + 2 \omega^2 \cos[\theta[t]], \theta[0] == \theta_0\}, \theta[t], t]$

... Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

... Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

... Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

... General: Further output of Solve::ifun will be suppressed during this calculation.

Out[41]:=
$$\left\{ \left\{ \theta[t] \rightarrow -2 \text{JacobiAmplitude}\left[\frac{i t \omega \sqrt{-1 + \cos[\theta_0]}}{\sqrt{2}} - \text{EllipticF}\left[\frac{\theta_0}{2}, \text{Csc}\left[\frac{\theta_0}{2}\right]^2\right], \text{Csc}\left[\frac{\theta_0}{2}\right]^2\right]\right\}, \right. \\ \left. \left\{ \theta[t] \rightarrow 2 \text{JacobiAmplitude}\left[\frac{1}{2} i \left(\sqrt{2} t \omega \sqrt{-1 + \cos[\theta_0]} - 2 i \text{EllipticF}\left[\frac{\theta_0}{2}, \text{Csc}\left[\frac{\theta_0}{2}\right]^2\right]\right), \text{Csc}\left[\frac{\theta_0}{2}\right]^2\right]\right\} \right\}$$

The Syngé and Griffith solution starts with the introduction of half angles

In[42]:= $w4[8] = w4[7] /. \text{Cos}[x_] \rightarrow 1 - 2 \text{Sin}[x/2]^2 // \text{Expand}$

Out[42]:= $\theta'[t]^2 == 4 \omega^2 \text{Sin}\left[\frac{\theta_0}{2}\right]^2 - 4 \omega^2 \text{Sin}\left[\frac{\theta[t]}{2}\right]^2$

Make the following change of variable.

In[43]:= $\text{def}[\phi] = \text{Sin}\left[\frac{\theta[t]}{2}\right] == \text{Sin}\left[\frac{\theta_0}{2}\right] \text{Sin}[\phi[t]]$

Out[43]:= $\text{Sin}\left[\frac{\theta[t]}{2}\right] == \text{Sin}\left[\frac{\theta_0}{2}\right] \text{Sin}[\phi[t]]$

In[44]:= **w4[9] = w4[8] /. (def[φ] /. Equal → Rule)**

Out[44]=
$$\theta' [t]^2 == 4 \omega^2 \sin\left[\frac{\theta_0}{2}\right]^2 - 4 \omega^2 \sin\left[\frac{\theta_0}{2}\right]^2 \sin[\phi[t]]^2$$

In[45]:= **w4[10] = Simplify /@ w4[9]**

Out[45]=
$$\theta' [t]^2 == 4 \omega^2 \cos[\phi[t]]^2 \sin\left[\frac{\theta_0}{2}\right]^2$$

Some side calculations are performed with the objective of expressing the lhs of this equation in terms of ϕ . Start by calculating an expression for $\phi' [t]$

In[46]:= **w4[11] = D[#, t] & /@ def[φ]**

Out[46]=
$$\frac{1}{2} \cos\left[\frac{\theta[t]}{2}\right] \theta' [t] == \cos[\phi[t]] \sin\left[\frac{\theta_0}{2}\right] \phi' [t]$$

In[47]:= **w4[12] = w4[11] /. Cos[$\frac{\theta[t]}{2}$] → $\sqrt{1 - \sin\left[\frac{\theta[t]}{2}\right]^2}$ /. (def[φ] /. Equal → Rule)**

Out[47]=
$$\frac{1}{2} \sqrt{\left(1 - \sin\left[\frac{\theta_0}{2}\right]^2 \sin[\phi[t]]^2\right)} \theta' [t] == \cos[\phi[t]] \sin\left[\frac{\theta_0}{2}\right] \phi' [t]$$

or

In[48]:= **w4[13] = Sol[w4[12], $\theta' [t]$]**

Out[48]=
$$\theta' [t] \rightarrow \frac{2 \cos[\phi[t]] \sin\left[\frac{\theta_0}{2}\right] \phi' [t]}{\sqrt{1 - \sin\left[\frac{\theta_0}{2}\right]^2 \sin[\phi[t]]^2}}$$

Now, $\theta(t)$ can be eliminated

In[49]:= **w4[14] = w4[10] /. w4[13]**

Out[49]=
$$\left(4 \cos[\phi[t]]^2 \sin\left[\frac{\theta_0}{2}\right]^2 \phi' [t]^2\right) / \left(1 - \sin\left[\frac{\theta_0}{2}\right]^2 \sin[\phi[t]]^2\right) == 4 \omega^2 \cos[\phi[t]]^2 \sin\left[\frac{\theta_0}{2}\right]^2$$

and solve for $\phi' (t)^2$.

In[50]:= **w4[15] = Solve[w4[14] /. $\phi' [t]^2 \rightarrow x$, x][[1, 1]] /. x → $\phi' [t]^2$**

Out[50]=
$$\phi' [t]^2 \rightarrow -\omega^2 \left(-1 + \sin\left[\frac{\theta_0}{2}\right]^2 \sin[\phi[t]]^2\right)$$

This particular ode can be conveniently solved by Mathematica. I will come back to that solution but, for the moment, I continue the historical treatment of Synge and Griffith.

Introduce the constant k

In[51]:= **def[k] = k == Sin[$\frac{\theta_0}{2}$]**

Out[51]= $k == \text{Sin}\left[\frac{\theta_0}{2}\right]$

and obtain

In[52]:= **w4[16] = w4[15] /. (def[k] // Reverse // ER) // RE**

Out[52]= $\phi'[t]^2 == -\omega^2 (-1 + k^2 \text{Sin}[\phi[t]]^2)$

Make a further change of variable

In[53]:= **def[y] = y[t] == Sin[$\phi[t]$]**

Out[53]= $y[t] == \text{Sin}[\phi[t]]$

After some manipulations

In[54]:= **w4[17] = D[#, t] & /@ def[y]**

Out[54]= $y'[t] == \text{Cos}[\phi[t]] \phi'[t]$

In[55]:= **w4[18] = w4[17] /. Cos[$\phi[t]$] $\rightarrow \sqrt{1 - \text{Sin}[\phi[t]]^2}$ /. (def[y] // Reverse // ER)**

Out[55]= $y'[t] == \sqrt{1 - y[t]^2} \phi'[t]$

In[56]:= **w4[19] = Solve[w4[18], $\phi'[t]$][[1, 1]]**

Out[56]= $\phi'[t] \rightarrow \frac{y'[t]}{\sqrt{1 - y[t]^2}}$

In[57]:= **w4[20] = w4[16] /. (def[y] // Reverse // ER) /. w4[19]**

Out[57]= $\frac{y'[t]^2}{1 - y[t]^2} == -\omega^2 (-1 + k^2 y[t]^2)$

the following ode for y is obtained.

In[58]:= **w4[21] = Solve[w4[20] /. $y'[t]^2 \rightarrow x$, x][[1, 1]] /. x $\rightarrow y'[t]^2$ // RE**

Out[58]= $y'[t]^2 == \omega^2 (-1 + y[t]^2) (-1 + k^2 y[t]^2)$

The square root of this is a defining ordinary differential equation for Jacobi elliptic functions
https://en.wikipedia.org/wiki/Jacobi_elliptic_functions

This ode can be solved but the results are again messy.

```
In[59]:= DSolve[{y'[t]^2 == ω^2 (-1 + y[t]^2) (-1 + k^2 y[t]^2), y[0] == y0}, y, t]
Out[59]:= {{y -> Function[{t}, InverseFunction[
  (EllipticF[ArcSin[#1], k^2] Sqrt[1 - #1^2] Sqrt[1 - k^2 #1^2]) / (Sqrt[-1 + #1^2] Sqrt[-1 + k^2 #1^2]) &] [
  -t ω + (Sqrt[1 - y0^2] Sqrt[1 - k^2 y0^2] EllipticF[ArcSin[y0], k^2]) / (Sqrt[-1 + y0^2] Sqrt[-1 + k^2 y0^2]) ]]],
  {y -> Function[{t}, InverseFunction[
  (EllipticF[ArcSin[#1], k^2] Sqrt[1 - #1^2] Sqrt[1 - k^2 #1^2]) / (Sqrt[-1 + #1^2] Sqrt[-1 + k^2 #1^2]) &] [
  t ω + (Sqrt[1 - y0^2] Sqrt[1 - k^2 y0^2] EllipticF[ArcSin[y0], k^2]) / (Sqrt[-1 + y0^2] Sqrt[-1 + k^2 y0^2]) ]]]}}
```

To make progress, I return and solve the differential equation for ϕ

```
In[60]:= DSolve[{φ'[t]^2 == -ω^2 (-1 + k^2 Sin[φ[t]]^2), φ[0] == φ0}, φ[t], t]
... Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.
... Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.
... Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.
... General: Further output of Solve::ifun will be suppressed during this calculation.
Out[60]:= {{φ[t] -> -JacobiAmplitude[t ω - EllipticF[φ0, k^2], k^2]},
  {φ[t] -> JacobiAmplitude[t ω + EllipticF[φ0, k^2], k^2]}}
```

This result is considerably cleaner than the results from solving the ode for θ or for y .

Using this result I can plot the motion of the pendulum. First, I solve for θ .

```
In[61]:= w4[22] = Solve[def[φ], θ[t]] /. C[1] -> θ
Out[61]:= {{θ[t] -> 2 (π - ArcSin[Sin[θ0/2] Sin[φ[t]])]}, {θ[t] -> 2 ArcSin[Sin[θ0/2] Sin[φ[t]]]}}
```

Work with the second branch

```
In[62]:= w4[23] = w4[22][[2, 1]] /. Sin[θ0/2] -> k
Out[62]:= θ[t] -> 2 ArcSin[k Sin[φ[t]]]
```

The initial condition is

In[63]:=

$$w4[24] = \text{Sin}\left[\frac{\theta[t]}{2}\right] == \text{Sin}\left[\frac{\theta_0}{2}\right] \text{Sin}[\phi[t]] \quad /. \quad t \rightarrow \theta \quad /. \quad \{\theta[0] \rightarrow \theta_0, \phi[0] \rightarrow \phi_0\}$$

Out[63]=

$$\text{Sin}\left[\frac{\theta_0}{2}\right] == \text{Sin}\left[\frac{\theta_0}{2}\right] \text{Sin}[\phi_0]$$

In[64]:=

$$w4[25] = \text{Solve}[w4[24], \phi_0][[1, 1]] \quad /. \quad C[1] \rightarrow \theta$$

Out[64]=

$$\phi_0 \rightarrow \pi - \text{ArcSin}\left[\text{Csc}\left[\frac{\theta_0}{2}\right] \text{Sin}\left[\frac{\theta_0}{2}\right]\right]$$

Numerical comparison of linear and nonlinear cases. The initial deflection of the pendulum from vertical is $\pi/4$


In[65]=

```

Module[{ $\omega = 1$ ,  $\theta_0 = \pi/4$ ,  $\phi_0 = \pi/2$ , tMax = 10, k,  $\psi_0$ ,  $\Phi$ ,  $\Theta$ ,  $\Theta_{\text{linear}}$ , lab, g},
  k = Sin[ $\frac{\theta_0}{2}$ ];
   $\psi_0$  = EllipticF[ $\phi_0$ , k2];
   $\Phi[t_]$  := JacobiAmplitude[t  $\omega$  +  $\psi_0$ , k2];
   $\Theta[t_]$  := 2 ArcSin[k Sin[ $\Phi[t]$ ]];
   $\Theta_{\text{linear}}[t_]$  :=  $\theta_0$  Cos[ $\omega t$ ];
  lab = Stl@StringForm["Solution of\n`",
    TraditionalForm[ $\phi'(t)^2 = -\omega^2 (-1 + k^2 \sin^2(\phi(t)))$ ]];
  g[1] = Plot[ $\Phi[t]$ , {t, 0, tMax}, PlotStyle -> Black, AxesLabel -> {Stl["t"], Stl[" $\phi$ "]},
    PlotLabel -> lab, PlotRange -> All, ImageSize -> 350];
  lab = Stl@StringForm["Nonlinear vs linear pendulum\n $\omega =$ ` `  $\theta_0 =$ ` `",  $\omega$ ,  $\theta_0$ ];
  g[2] =
    Plot[{ $\Theta[t]$ ,  $\Theta_{\text{linear}}[t]$ }, {t, 0, tMax}, PlotStyle -> {Black, Directive[Black, Dashed]},
      PlotLegends -> Placed[{"Nonlinear", "Linear"}, Right], AxesLabel ->
        {Stl["t"], Stl[" $\theta$ "]}, PlotLabel -> lab, PlotRange -> All, ImageSize -> 350];
  Grid[{{g[1]}, {g[2]}}]]

```

Out[65]=



Note that the period of the general nonlinear pendulum is longer than the period calculated under the

linear approximation in Section 1. In the next section I will calculate the dependence of the period T on the amplitude of oscillations.

5 Calculation of the dependence of the period on oscillation amplitude

I again work through a historical treatment of the classical pendulum — calculating how the period of the pendulum is affected by the finite amplitude of oscillations.

From Section 4

In[66]:= $w5[1] = \theta'[t]^2 == -2 \omega^2 \text{Cos}[\theta_0] + 2 \omega^2 \text{Cos}[\theta[t]]$

Out[66]:= $\theta'[t]^2 == -2 \omega^2 \text{Cos}[\theta_0] + 2 \omega^2 \text{Cos}[\theta[t]]$

In[67]:= $w5[2] = \text{Solve}[w5[1], \theta'[t]]$

Out[67]:= $\left\{ \left\{ \theta'[t] \rightarrow -\sqrt{2} \omega \sqrt{-\text{Cos}[\theta_0] + \text{Cos}[\theta[t]]} \right\}, \left\{ \theta'[t] \rightarrow \sqrt{2} \omega \sqrt{-\text{Cos}[\theta_0] + \text{Cos}[\theta[t]]} \right\} \right\}$

Work with the positive branch

In[68]:= $w5[3] = \text{Simplify}[w5[2][[2, 1]], \text{Assumptions} \rightarrow \{\omega > 0\}] // \text{RE}$

Out[68]:= $\theta'[t] == \sqrt{2} \omega \sqrt{-\text{Cos}[\theta_0] + \text{Cos}[\theta[t]]}$

Proceeding manually

In[69]:= $w5[4] = w5[3] /. \theta'[t] \rightarrow d\theta/dt // \text{RE}$

Out[69]:= $\frac{d\theta}{dt} == \sqrt{2} \omega \sqrt{-\text{Cos}[\theta_0] + \text{Cos}[\theta[t]]}$

In[70]:= $w5[5] = \text{Solve}[w5[4], dt][[1, 1]] // \text{RE}$

Out[70]:= $dt == d\theta / \left(\sqrt{2} \omega \sqrt{-\text{Cos}[\theta_0] + \text{Cos}[\theta[t]]} \right)$

This equation is integrated. At t goes from 0 to $T/4$, θ goes from 0 to θ_0 , its initial deflection. I use the symbol *Int* rather than *Integrate* to suppress Mathematica's immediate attempt to value the integral.

$$\text{In[71]:= } w5[6] = \frac{T}{4} == \text{Int}[w5[5][[2]], \{\theta, \theta, \theta_0\}]$$

$$\text{Out[71]= } \frac{T}{4} == \text{Int}\left[\frac{d\theta}{\sqrt{2} \omega \sqrt{-\cos[\theta_0] + \cos[\theta[t]]}}\right], \{\theta, \theta, \theta_0\}]$$

Now comes a change of variable and some algebraic manipulations. The objective is to cast this expression into a form involving the elliptical integral of the first kind

$$F(k, x) = \int_0^x \frac{d\phi}{1 - k^2 \sin^2(\phi)}$$

$$\text{In[72]:= } w5[7] = w5[6] /. \theta[t] \rightarrow \theta /. \text{Cos}[x_] \rightarrow 1 - 2 \text{Sin}[x/2]^2$$

$$\text{Out[72]= } \frac{T}{4} == \text{Int}\left[\frac{d\theta}{\sqrt{2} \omega \sqrt{\left(-2 \text{Sin}\left[\frac{\theta}{2}\right]^2 + 2 \text{Sin}\left[\frac{\theta_0}{2}\right]^2\right)}}\right], \{\theta, \theta, \theta_0\}]$$

$$\text{In[73]:= } w5[8] = w5[7] /. (\text{def}[k] // \text{ER} // \text{Reverse})$$

$$\text{Out[73]= } \frac{T}{4} == \text{Int}\left[\frac{d\theta}{\sqrt{2} \omega \sqrt{2 k^2 - 2 \text{Sin}\left[\frac{\theta}{2}\right]^2}}\right], \{\theta, \theta, \theta_0\}]$$

A change of variables requires several manipulations

$$\text{In[74]:= } \text{def}[\phi] = \text{Sin}[\phi] == \frac{1}{k} \text{Sin}\left[\frac{\theta}{2}\right]$$

$$\text{Out[74]= } \text{Sin}[\phi] == \frac{\text{Sin}\left[\frac{\theta}{2}\right]}{k}$$

$$\text{In[75]:= } w5[9] = w5[8] /. \text{Sol}[\text{def}[\phi], \text{Sin}[\theta/2]]$$

$$\text{Out[75]= } \frac{T}{4} == \text{Int}\left[\frac{d\theta}{\sqrt{2} \omega \sqrt{2 k^2 - 2 k^2 \text{Sin}[\phi]^2}}\right], \{\theta, \theta, \theta_0\}]$$

This can be simplified

$$\text{In[76]:= } w5[10] = \text{Simplify}[w5[9], \{k > \theta\}] // \text{PowerExpand}$$

$$\text{Out[76]= } T == 4 \text{Int}\left[\frac{d\theta \text{Sec}[\phi]}{2 k \omega}\right], \{\theta, \theta, \theta_0\}]$$

An expression for $d\theta$ is also required

In[77]:= **w5[11] = D[#, θ] & /@ (def[ϕ] /. $\phi \rightarrow \phi[\theta]$)**

Out[77]=
$$\text{Cos}[\phi[\theta]] \phi'[\theta] == \frac{\text{Cos}\left[\frac{\theta}{2}\right]}{2k}$$

In[78]:= **w5[12] = w5[11] /. $\phi'[\theta] \rightarrow d\phi/d\theta$ /. $\phi[\theta] \rightarrow \phi$**

Out[78]=
$$\frac{d\phi \text{Cos}[\phi]}{d\theta} == \frac{\text{Cos}\left[\frac{\theta}{2}\right]}{2k}$$

In[79]:= **w5[13] = Sol[w5[12], $d\theta$] /. $\phi[\theta] \rightarrow \phi$**

Out[79]=
$$d\theta \rightarrow 2 d\phi k \text{Cos}[\phi] \text{Sec}\left[\frac{\theta}{2}\right]$$

Then

In[80]:= **w5[14] = w5[10] /. w5[13]**

Out[80]=
$$T == 4 \text{Int}\left[\frac{d\phi \text{Sec}\left[\frac{\theta}{2}\right]}{\omega}, \{\theta, \theta_0, \theta_0\}\right]$$

Yet more work is required to eliminate $\text{Sec}(\theta/2)$ in terms of ϕ

In[81]:= **w5[15] = Sol[def[ϕ], $\text{Sin}[\theta/2]$] // RE**

Out[81]=
$$\text{Sin}\left[\frac{\theta}{2}\right] == k \text{Sin}[\phi]$$

In[82]:= **w5[16] = w5[15] /. $\text{Sin}\left[\frac{\theta}{2}\right] \rightarrow \sqrt{1 - \text{Cos}\left[\frac{\theta}{2}\right]^2}$**

Out[82]=
$$\sqrt{1 - \text{Cos}\left[\frac{\theta}{2}\right]^2} == k \text{Sin}[\phi]$$

In[83]:= **w5[17] = Solve[w5[16], $\text{Cos}\left[\frac{\theta}{2}\right]$] [[2, 1]]**

Out[83]=
$$\text{Cos}\left[\frac{\theta}{2}\right] \rightarrow \sqrt{1 - k^2 \text{Sin}[\phi]^2}$$

In[84]:= **w5[18] = (1/#) & /@ w5[17]**

Out[84]=
$$\text{Sec}\left[\frac{\theta}{2}\right] \rightarrow \frac{1}{\sqrt{1 - k^2 \text{Sin}[\phi]^2}}$$

Using this result

In[85]:=

w5[19] = w5[14] /. w5[18]

Out[85]=

$$T = 4 \text{Int} \left[\frac{d\phi}{\omega \sqrt{1 - k^2 \text{Sin}[\phi]^2}}, \{\theta, \theta, \theta_0\} \right]$$

It remains to handle the limits of integration. Recall

In[86]:=

def[k]

Out[86]=

$$k = \text{Sin} \left[\frac{\theta_0}{2} \right]$$

In[87]:=

w5[20] = def[phi] /. Sol[def[k], k]

Out[87]=

$$\text{Sin}[\phi] = \text{Csc} \left[\frac{\theta_0}{2} \right] \text{Sin} \left[\frac{\theta}{2} \right]$$

At the lower limit $\theta = 0$

In[88]:=

w5[21] = w5[20] /. theta -> 0

Out[88]=

$$\text{Sin}[\phi] = 0$$

So, when $\theta = 0$ corresponds to $\phi = 0$

At the upper limit $\theta = \theta_0$

In[89]:=

w5[22] = w5[20] /. theta -> theta_0

Out[89]=

$$\text{Sin}[\phi] = 1$$

So when $\theta = \theta_0$, $\phi \rightarrow \pi/2$

This information is used to rewrite the limits of integration

In[90]:=

w5[23] = w5[19] /. {theta, theta, theta_0} -> {phi, 0, pi/2}

Out[90]=

$$T = 4 \text{Int} \left[\frac{d\phi}{\omega \sqrt{1 - k^2 \text{Sin}[\phi]^2}}, \left\{ \phi, 0, \frac{\pi}{2} \right\} \right]$$

Turn Mathematica loose

In[91]:=

w5[24] = w5[23] /. Int -> Integrate /. dphi -> 1

Out[91]=

$$\text{ConditionalExpression} \left[T = \frac{4 \text{EllipticK} \left[\frac{k^2}{-1+k^2} \right]}{\sqrt{1 - k^2} \omega}, \text{Re}[k^2] \leq 1 \mid \mid k^2 \notin \text{Reals} \right]$$

Here, EllipticK is $F(\frac{\pi}{2}, k)$.

Recall the definition of k

In[92]=

```
def[k]
```

Out[92]=

$$k = \text{Sin}\left[\frac{\theta_0}{2}\right]$$

Since the initial displacement of the pendulum must satisfy $\theta_0 \leq \pi$, we have $k < 1$

In[93]=

```
w5[25] = Simplify[w5[24], {k^2 <= 1}]
```

Out[93]=

$$T = \frac{4 \text{EllipticK}\left[\frac{-k^2}{-1+k^2}\right]}{\sqrt{1-k^2} \omega}$$

This can be further simplified by eliminating k

In[94]=

```
w5[26] = (# \omega / (2 \pi)) & /@ (w5[25] /. (def[k] // ER) // Simplify // PowerExpand)
```

Out[94]=

$$\frac{T \omega}{2 \pi} = \frac{1}{\pi} 2 \text{EllipticK}\left[-\text{Tan}\left[\frac{\theta_0}{2}\right]^2\right] \text{Sec}\left[\frac{\theta_0}{2}\right]$$

An expansion for θ_0 indicates the nature of the dependence of T on θ_0

In[95]=

```
w5[27] = w5[26] [[1]] == Normal@Series[w5[26] [[2]], {\theta_0, 0, 5}]
```

Out[95]=

$$\frac{T \omega}{2 \pi} = 1 + \frac{\theta_0^2}{16} + \frac{11 \theta_0^4}{3072}$$

In[96]=


```

Module[{lab, F, Fapprox},
  F[ $\theta_0$ _,  $\omega$ _] :=  $\frac{1}{\pi} 2 \text{EllipticK}[-\text{Tan}[\frac{\theta_0}{2}]^2] \text{Sec}[\frac{\theta_0}{2}]$ ;
  Fapprox[ $\theta_0$ _,  $\omega$ _] :=  $1 + \frac{\theta_0^2}{16} + \frac{11 \theta_0^4}{3072}$ ;

  lab = Stl@StringForm["Dependence of period on amplitude",];
  Plot[{F[ $\theta_0$ , 1], Fapprox[ $\theta_0$ , 1]},
    { $\theta_0$ , 0,  $\pi/2$ }, PlotStyle -> {Black, Directive[Black, Dashed]},
    PlotLegends -> Placed[{TraditionalForm[ $\frac{1}{\pi} 2 \text{EllipticK}[-\text{Tan}[\frac{\theta_0}{2}]^2] \text{Sec}[\frac{\theta_0}{2}]$ ],
      TraditionalForm[ $1 + \frac{\theta_0^2}{16} + \frac{11 \theta_0^4}{3072}$ ]}, Right],
    AxesLabel -> {Stl[" $\theta_0$ "], Stl[" $\frac{T \omega}{2 \pi}$ ]"}, PlotLabel -> lab, PlotRange -> All ] ]

```

Out[96]=




Check of analytical $T(\theta_0)$ against numerical calculation

After such a long derivation/calculation, it is a good idea to compare results with a numerical solution of the original equation.

In[97]:=

```
Module[{ $\omega = 1$ , tMax = 10,  $\theta_0 = \pi/4$ , soln, solnLinear, lab},
  soln =
    NDSolve[{ $\omega^2 \text{Sin}[\theta[t]] + \theta''[t] == 0$ ,  $\theta[0] == \theta_0$ ,  $\theta'[0] == 0$ },  $\theta[t]$ , {t,  $\theta$ , tMax}];
  solnLinear = NDSolve[{ $\omega^2 \theta[t] + \theta''[t] == 0$ ,  $\theta[0] == \theta_0$ ,  $\theta'[0] == 0$ },
     $\theta[t]$ , {t,  $\theta$ , tMax}];
  lab = Stl@StringForm[" $\omega = \`$    $\theta_0 = \`$ \nComparing linear and nonlinear solutions",
     $\omega$ ,  $\theta_0$ ];
  Plot[{Evaluate[ $\theta[t]$  /. soln], Evaluate[ $\theta[t]$  /. solnLinear]}, {t,  $\theta$ , tMax},
    PlotRange -> All, PlotStyle -> {Black, Directive[Black, Dashed]},
    AxesLabel -> {Stl["t"], Stl[" $\theta$ "]}, PlotLabel -> lab,
    PlotLegends -> Placed[{" $\omega^2 \text{Sin}[\theta[t]] + \theta''[t] == 0$ ", " $\omega^2 \theta[t] + \theta''[t] == 0$ "}, Right]]]
```

Out[97]=




This relation between linear and nonlinear numerical solutions is the same as was seen for analytical solutions above.

I want to compare the dependence of period on oscillation amplitude. A numerical value for the period can be calculated using *WhenEvent* to stop the integration when $\theta'(T) < 0$ for the second time.

In[98]=

```
Module[{ $\omega = 1$ , tMax = 10,  $\theta_0 = \pi/8$ , soln, T, lab},
  soln = NDSolveValue[{ $\omega^2 \text{Sin}[\theta[t]] + \theta''[t] == 0$ ,  $\theta[0] == \theta_0$ ,
     $\theta'[0] == 0$ , WhenEvent[ $\theta'[t] < 0$ , {T = t, "StopIntegration"}},
    "LocationMethod" → "LinearInterpolation"}],  $\theta$ , {t, 0, tMax}];
  lab = Stl@StringForm[" $\omega = \` \backslash$ nCalculated period T = `",  $\omega$ , T];
  Plot[{soln[t], soln'[t]}, {t, 0, T}, PlotLabel → lab,
  AxesLabel → {Stl["t"], Stl[" $\theta$ "]}, PlotStyle → {Black, Blue},
  PlotLegends → Placed[{" $\theta(t)$ ", " $\theta'(t)$ "}, Right]]]
```

Out[98]=



FindPeriod returns the numerically determined period T .

In[99]=


```
Clear[FindNumericalPeriod];
FindNumericalPeriod[ $\omega_$ ,  $\theta_0_$ , tMax_] :=
Module[{soln, T, lab},
  soln = NDSolveValue[{ $\omega^2 \text{Sin}[\theta[t]] + \theta''[t] == 0$ ,  $\theta[0] == \theta_0$ ,
     $\theta'[0] == 0$ , WhenEvent[ $\theta'[t] < 0$ , {T = t, "StopIntegration"}},
    "LocationMethod" → "LinearInterpolation"}],  $\theta$ , {t, 0, tMax}];
  T]
```

I check the analytical formula for $T(\theta_0)$ against a numerical calculation. Agreement is excellent.

In[101]=

```
Module[{ $\omega = 1$ , tMax = 10,  $\theta_0 = \pi/4$ , T, results, lab},
  T[ $\theta_0$ ] := Module[{k = Sin[ $\frac{\theta_0}{2}$ ]},  $\frac{4 \text{EllipticK}[\frac{k^2}{-1+k^2}]}{\sqrt{1-k^2}} \omega$ ];
  results =
    Table[{ $\theta_0$ , FindNumericalPeriod[ $\omega$ ,  $\theta_0$ , tMax]}, { $\theta_0$ ,  $\pi/20$ ,  $\pi/2$ ,  $\pi/20$ ]];
  lab = Stl@StringForm["Dependence of period on amplitude  $\omega =$ 
    ``\nAnalytical (black line), numerical (black circles)",  $\omega$ ];
  Plot[T[ $\theta_0$ ], { $\theta_0$ ,  $\pi/20$ ,  $\pi/2$ }, PlotLabel -> lab, AxesLabel -> {Stl[" $\theta_0$ "], Stl["T"]},
    PlotStyle -> {Black}, Epilog -> {OC[#, Black] &/@results},
    Ticks -> {Table[i  $\pi/8$ , {i, 0, 4}], Automatic},
    PlotRange -> {{0,  $\pi/2 + 0.1$ }, Automatic}]]
```

Out[101]=



Graphics

Generation of the initial figure

In[102]=

```
Clear[GVector];
GVector[tail_, tip_, label_, posLabel_, size_, color_] :=
  {color, Arrowheads[size], Arrow[{tail, tip}], Text[Stl[label], posLabel]}
```

In[104]=

```

Module[{offset = 0.1,  $\theta_P = 7\pi/8$ , L = 1, O, X, Z, P, pendulum, xAxis,
  zAxis, axes, background,  $\theta$ arc, mgVec, maVec, TVec, vectors, OText, PtoC},
  PtoC[r_,  $\theta$ _] := r {Sin[ $\theta$ ], Cos[ $\theta$ ]};
  {O, X, Z} = {{0, 0}, {1, 0}, {0, 1}};
  P = PtoC[L,  $\theta_P$ ];
  background = {Line[1.1 {-X, X}], Line[1.5 {-Z, {0, 0}}]};
   $\theta$ arc = {Arrowheads[Small], Arrow@Table[PtoC[ $\frac{L}{3}$ ,  $\theta$ ], { $\theta$ ,  $\pi$ ,  $\theta_P$ ,  $-\frac{\theta_P}{64}$ }],
    {Black, Text[Stl[" $\theta$ "], PtoC[1.2  $\frac{L}{3}$ ,  $\frac{\pi + \theta_P}{2}$ ]]}};
  pendulum = {Directive[Brown, Thickness[0.015]], Line[{O, P}], {Black,
    PointSize[0.03], Point[P]}, {Brown, Text[Stl["L"], P/2 + {offset, 0}]}];
  mgVec = GVector[P, P + {0, -0.3}, "m  $\vec{g}$ ", P + {0, -0.4}, Medium, Black];
  maVec = GVector[P, P - 0.3 {-Cos[ $\theta_P$ ], Sin[ $\theta_P$ ]},
    "m  $\vec{a}$ ", P + {-0.4, -offset}, Medium, Black];
  TVec = GVector[P, 0.6 P, " $\vec{T}$ ", 0.6 P + {-0.1, -0.1}, Medium, Black];
  axes = With[{origin = {0.5, -0.5}},
    xAxis = GVector[origin, origin + {0.1, 0}, "x", origin + {0.15, 0}, Small, Gray];
    zAxis = GVector[origin, origin + {0, 0.1}, "z", origin + {0, 0.15}, Small, Gray];
    {xAxis, zAxis}];
  vectors = {mgVec, maVec, TVec, axes};
  OText = Text["O", O + offset {-1, -1}];
  Graphics[{{GrayLevel[0.7], background,  $\theta$ arc}, pendulum, vectors, OText},
    Axes  $\rightarrow$  None, PlotLabel  $\rightarrow$  Stl["Forces on a displaced pendulum"]]

```

Out[104]=

